AD-A189 136 AVSCOM'S MODIFICATIONS TO TELEDYNE SYSTEMS COMPANY'S 1/1
AIR-TO-AIR FIRE CONT..(U) ARMY AVIATION SYSTEMS COMMAND
ST LOUIS MO D J BREYER NOV 87 USAAVSCOM-TM-87-F-3

UNCLASSIFIED F/G 19/5 NL




1.0
2.8
2.5
3.2
2.2
3.6
4.0
2.0
1.1
1.8
1.25
1.4
1.6

2

# USAAVSCOM

Technical Memorandum
TM 87-F-3

AD



# AVSCOM'S MODIFICATIONS TO TELEDYNE SYSTEMS COMPANY'S AIR-TO-AIR FIRE CONTROL SYSTEM SIMULATION MODEL

D.J. BREYER
Operations Research Analyst

NOVEMBER 1987

Approved for public release;
distribution is unlimited.



U.S. ARMY
AVIATION
SYSTEMS COMMAND

## DISCLAIMER

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

# REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

| | | |
|---|---|---|
| 1. REPORT NUMBER<br>TM-87-F-3 | 2. GOVT ACCESSION NO.<br>AD-A187 123 | 3. RECIPIENT'S CATALOG NUMBER |
| 4. TITLE (and Subtitle)<br>AVSCOM's Modifications to Teledyne Systems Company's Air-to-Air Fire Control System Simulation Model | | 5. TYPE OF REPORT & PERIOD COVERED<br><br>6. PERFORMING ORG. REPORT NUMBER |
| 7. AUTHOR(s)<br>Daniel J. Breyer | | 8. CONTRACT OR GRANT NUMBER(s) |
| 9. PERFORMING ORGANIZATION NAME AND ADDRESS<br>U.S. Army Aviation Systems Command (AVSCOM)<br>Developmental Systems Analysis Division(AMSAV-BA)<br>4300 Goodfellow Boulevard<br>St. Louis, MO 63120-1798 | | 10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS |
| 11. CONTROLLING OFFICE NAME AND ADDRESS<br>U.S. Army Aviation Systems Command<br>Directorate for Systems & Cost Analysis (AMSAV-B)<br>4300 Goodfellow Boulevard<br>St. Louis, MO 63120-1798 | | 12. REPORT DATE<br>November 1987<br>13. NUMBER OF PAGES<br>24 |
| 14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office) | | 15. SECURITY CLASS. (of this report)<br>Unclassified<br>15a. DECLASSIFICATION/DOWNGRADING SCHEDULE |

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution is unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Air-to-Air
Fire Control Equations

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This report describes changes made by AVSCOM's Developmental Systems Analysis Division to Teledyne Systems Company's ARTOAR simulation model. The changes are discussed in length and the coding required for each change is also provided. Applications of the changes to specific projects are briefly discussed when appropriate.

DD FORM 1473 1 JAN 73 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

 SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

## ACKNOWLEDGEMENTS

The author wishes to express his appreciation to Dr. Anthony Kassos and Mr. Arnold Arconati for their very capable and always illuminating insights into the development of this report.

TABLE OF CONTENTS

## LIST OF TABLES



## LIST OF FIGURES

## INTRODUCTION

This report has been written to document the changes made to Teledyne Systems Company's (TSC) Attack Helicopter Air-to-Air (ATA) Fire Control System Simulation model (AIRTOAIR). The model was acquired by the Developmental Systems Analysis Division of the U.S. Army Aviation Systems Command (USAAVSCOM) in April of 1986 under the direction of the Directorate for Engineering. The original model was developed under contract from the USAAVSCOM's Aviation Applied Technology Directorate (AATD) in Fort Eustis, Virginia, and the U.S. Army Armament Research, Development and Engineering Center (ARDEC) in Dover, New Jersey.

## II. BACKGROUND TO THE MODEL

The AIRTOAIR Simulation was used as a tool by TSC to assess the effectiveness of several mechanizations of fire control equations. These equations were being formulated with the intent of becoming a product improvement to the existing attack helicopters, the AH-1S COBRA and the AH-64 APACHE. The model simulates a one-on-one, non-dueling, close-in combat scenario using the automatic cannon. Since it is a fire control system representation, all of the sensors that feed information to the fire control computer (FCC) are

modelled. These include:

1. Laser rangefinder.

2. Sight-line measurement via gimbals.

3. Sight-line rates via gyroscopes.

4. Air data sensors.

5. Doppler radar.

6. Heading attitude reference system (APACHE).

7. Vertical gyro/magnetic compass (COBRA).

Error models consisting of instrument biases, scale factors and random noise are used to simulate these sensors. They are modularized so that additions or alterations can easily be made if desired. This allows the analyst to not only study the APACHE or COBRA with ATA equations, but to simulate futuristic aircraft with new sensors, such as a millimeter wave radar or muzzle velocity sensor.

The fire control equations consist of three interrelated modules, a filtering algorithm used to estimate the target states, a prediction algorithm to extrapolate these target states one bullet time of flight into the future, and an airborne ballistics algorithm to determine the gun-laying vectors to put the bullets on the predicted future target position. Two filters were in the original model--a seven-state, rotating frame Kalman filter and a variable coefficient alpha-beta-gamma filter. A quadratic prediction algorithm was employed, and the airborne ballistics were those developed by Mr. Harold Breaux of the USA Ballistics Research Laboratory (BRL).[1]

The model then implements a modified 4-degree-of-freedom (DOF) real world

ballistic algorithm (developed by Mr. Tom Hutchings of ARDEC) to simulate bullet flyout to the target and computes miss distances and burst probability of kill. All of this background information is explained in much greater detail within the final report that TSC delivered to ARDEC under contract specifications.[2] An overall flowchart of the model, taken from reference 2, is shown on the following page (Figure 1). The model is programmed in FORTRAN 77 and is currently running on a DEC VAX (at TSC), a CDC (at ARDEC) and an IBM 3781 (at AVSCOM).

## III. THE CHANGES

Since acquiring the model, many modifications have been made. Some have been as easy as adding several lines of coding, while others have required new input data files or subroutines. The changes in coding will be provided, along with a description of the methodology. It should be noted here that the model continues to be modified, and future alterations will need to be explained under another cover, although several planned ones will be touched upon here.

The modifications to be discussed include the following:

1. 20MM M70 Pyrotechnically-initiated explosive (PIE) round added.
2. End-game methodology improved.
3. Existing APACHE and COBRA target tracking Kalman filters added.
4. Linear prediction algorithm added.
5. Change from deterministic to Monte Carlo with appropriate statistics computed.




"AIRTOAIR" PROGRAM FLOWCHART

Figure 1

6. Graphic program accessed to study Kalman filter performance.

7. More engagement files created and run.

Since the purpose of this report is to explain the modifications, applications of these to specific projects will be discussed only briefly whenever applicable.

A. Addition of the 20MM PIE round included both "real world" data and airborne ballistic coefficients. The data needed by the 4-DOF algorithm is produced by BRL from actual firings of the rounds (Table 1). The airborne ballistics algorithm in the fire control computer uses 24 different coefficients, which are derived from fitted approximations to the raw data, to closely predict bullet time-of-flight to the estimated target position (Table 2). Lethality data from the round against various red air and ground targets also needed to be obtained from BRL for the end game calculations.

The 20MM PIE has been a leading candidate for the LHX cannon, primarily due to its aerodynamic qualities, resulting in decreased time-of-flight, and its excellent destructive capabilities on thin-skinned airborne targets. Numerous trade-off studies between the 20MM PIE and 30MM HEDP (high explosive/dual purpose) have been conducted for LHX purposes to examine the time-of-flight versus lethality issue. The studies show that the burst probability of kill is highly engagement-dependent, and that normally more 20MM rounds will strike the target than 30MM, but sometimes probability of kill is greater for the 30MM due to its exceptional destructive ability.

B. The most important and time-consuming change was the reprogramming of the end game methodology. In the original model, target vulnerable area

20MM PGU-28/B "REAL WORLD" COEFFICIENTS

| | |
|---|---|
| LIFT FACTOR 1.0 | |
| YAW DRAG FACTOR | 1.0 |
| DOWN RANGE YAW LIMIT CYCLE (DEGREES) | 6.0 |
| PROJECTILE DIAMETER (METERS) | .0199 |
| PROJECTILE MASS (KG) | .1015 |
| MUZZLE VELOCITY (M/SEC) | 1051.56 |
| INITIAL SPIN (RAD/SEC) | 13071.55 |
| CENTER OF GRAVITY (CALIBER) | 2.8846 |
| LONGITUDINAL MOMENT OF INERTIA (KG*M**2) | .00000547 |
| TRANSVERSE MOMENT OF INERTIA (KG*M**2) | .00003498 |
| MAGNUS FORCE POSITION (CALIBER) | 0.0 |
| DRAG FUNCTION FORM FACTOR | 1.0 |
| FIN CANT ANGLE (DEGREES) | 0.0 |

Table 1

## 20MM PGU-28/B "REAL WORLD" COEFFICIENTS

| MACH NUMBER | DRAG | YAW DRAG | LIFT FORCE | SPIN DAMPING MOMENT | OVERTURNING MOMENT | MAGNUS MOMENT |
|---|---|---|---|---|---|---|
| 0.0 | .160 | 3.05 | 1.40 | -.021 | 3.54 | .135 |
| 0.2 | | | | | 3.60 | |
| 0.3 | .160 | 3.05 | | -.021 | | |
| 0.4 | | | | -.0209 | | |
| 0.5 | .161 | 3.20 | | -.0206 | | |
| 0.6 | | | | -.0204 | 3.79 | |
| 0.7 | .165 | 3.70 | | | | |
| 0.75 | | | | -.0195 | | |
| 0.8 | .170 | 4.20 | | | 3.94 | |
| 0.85 | .175 | 4.50 | | | 3.97 | |
| 0.9 | .190 | 5.20 | | -.0186 | 4.04 | |
| 0.95 | .251 | | 1.40 | | 4.10 | |
| 1.0 | .363 | 7.10 | | -.0180 | 3.99 | |
| 1.05 | .423 | 7.55 | | | | |
| 1.1 | .430 | 7.80 | 1.57 | -.0174 | 3.78 | |
| 1.15 | .432 | | | | | |
| 1.2 | .428 | 8.10 | | | | |
| 1.25 | | 8.20 | | | | |
| 1.3 | .417 | 8.10 | 1.78 | | 3.52 | |
| 1.4 | | 7.90 | | -.0160 | | |
| 1.45 | | | | | 3.39 | |
| 1.5 | .393 | 7.65 | 1.90 | | | |
| 1.6 | | | | -.0152 | | |
| 1.7 | | | 2.10 | | | |
| 1.8 | .365 | | 2.16 | -.0144 | 3.21 | |
| 2.0 | | | 2.26 | -.0137 | 3.14 | |
| 2.2 | | | 2.34 | -.0131 | | |
| 2.4 | | | 2.38 | -.0126 | 3.00 | |
| 2.5 | .302 | | | | | |
| 2.6 | | | 2.42 | -.0120 | | |
| 2.8 | | 3.10 | 2.43 | | | |
| 2.85 | | | | | 2.87 | |
| 2.9 | .278 | | | | | |
| 3.0 | | 2.40 | 2.42 | -.0110 | 2.85 | |
| 3.2 | .261 | 1.90 | | -.0106 | | |
| 3.25 | | | | | 2.79 | |
| 3.3 | | 1.70 | | | | |
| 3.5 | .245 | 1.50 | 2.42 | -.0100 | 2.75 | .135 |

Table 1
(Continued)

20MM PGU-28/B AIRBORNE BALLISTIC COEFFICIENTS

| ARRAY ELEMENT | DEFINITION | COEFFICIENT |
|---|---|---|
| CA(1,J) | Drag at Standard Density | .00050617 |
| CA(2,J) | Initial Mach Number | 3.09 |
| CA(3,J) | Component of Drag Coefficient | 8.1648 |
| CA(4,J) | Muzzle Velocity | 1051.6 |
| CA(5,J) | Time of Flight | .99313 |
| CA(6,J) | Time of Flight | -2.9300 |
| CA(7,J) | Time of Flight | .42296 |
| CA(8,J) | Time of Flight | 5.2728 |
| CA(9,J) | Time of Flight | -.22338 |
| CA(10,J) | Time of Flight | -.14626 |
| CA(11,J) | Time of Flight | .55981 |
| CA(12,J) | Time of Flight | 0.0 |
| CA(13,J) | Direction Cosines | -.33719 |
| CA(14,J) | Direction Cosines | -1.0 |
| CA(15,J) | Direction Cosines | -.41969 |
| CA(16,J) | Direction Cosines | -.24146 |
| CA(17,J) | Direction Cosines | .32397 |
| CA(18,J) | Burnout Velocity | 0.0 |
| CA(19,J) | Burnout Velocity | 0.0 |
| CC(1,J) | Lift Moment Coefficient | 2.42 |
| CC(2,J) | Overturning Moment Coefficient | 2.83 |
| CC(3,J) | Lift Constant (Dimensionless) | 0.0 |
| CC(4,J) | Overturning Constant (Dimensionless) | 0.0 |
| CC(5,J) | Aerodynamic Jump Constant (Dimensionless) | .033378 |

Table 2

was input in square meters and then reduced to a circle. The area was capable of being changed during the engagement, but TSC only slightly varied it based on severity of the maneuvers being modelled. After a miss distance was calculated to the actual target position for each bullet, it was compared to the radius of the effective area circle to decide whether the bullet hit or missed. If scored a hit, the burst probability of kill was increased using a constant single shot probability of kill given hit for each type of round modelled (20MM or 30MM).

The new methodology is based upon the typical end game used by the U.S. Army Materiel Systems Analysis Activity (AMSAA) within their simulation models. Vulnerable and presented areas are input, in square meters, using front, rear, bottom, top and two sides. The vulnerable areas are dependent upon striking velocity of the projectile and are further delineated by damage conditions, i.e., mission abort, forced landing and attrition kill. Logic has been programmed in to use either the shoebox or Armitage ellipsoid representation for the target, and the kill definition can also be chosen based on the above damage conditions.

The second part of the new end game is used when the bullet reaches its closest approach to the actual target position. A target coordinate system had to be defined so that target and projectile velocities could be computed. The three faces of the target that are exposed to the incoming round are determined and the relative projectile striking velocity, as well as vulnerable areas based on that striking velocity, are computed. These vulnerable areas, along with the presented area, are combined into one area for each damage

condition using either target representation mentioned above (shoebox or ellipsoid). A presented area radius is computed in the same fashion as the TSC methodology and compared to the miss distance. If that bullet is scored a hit, a probability of kill given hit (ratio of vulnerable area to presented area) is calculated dependent upon kill criteria used and the burst probability of kill is augmented appropriately. The main difference between the two methodologies is that the new one allows for the vulnerable and presented areas, as well as the single shot probability of kill given hit, to vary based upon engagement geometry and each individual bullet's flight path. The subroutines integrated into the model that handle the new end game methodology are TRGATD (determination of exposed areas) and TRGINP (inputs the new vulnerability data), which are shown in Figure 2. Additions were, made to subroutines MISDIS and PROBKL for the new methodology.

C. The next most important modification was the addition of the existing APACHE and COBRA Kalman tracking filters to the model. The first step involved requesting the algorithms directly from McDonnell Douglas Helicopter Company (APACHE) and Bell Helicopter (COBRA). After clarifying questions on variables and sources of data for the filters, the two algorithms were integrated into the model.

The APACHE currently utilizes a 3-state Kalman filter which estimates target range, range rate and acceleration along the line-of-sight. Additional measurements needed by this filter, beyond the ones used by the TSC-developed 7-state Kalman filter, are provided to the FCC by the APACHE's Heading Attitude

```
C
C****************************************************************************
C
C         NO. 60
C
C  THE CALL TO THIS SUBROUTINE FROM 'MISDIS' IS WHEN THE BULLET HAS      *
C  REACHED ITS CLOSEST APPROACH TO THE TARGET.   THE STRIKING            *
C  VELOCITY (VS), VULNERABLE AREA DEPENDANT UPON THAT STRIKING           *
C  VELOCITY, AND TARGET AND PROJECTILE VELOCITIES IN THE TARGET COORD-*
C  INATE SYSTEM ARE COMPUTED AT THIS TIME (POSITIONS ARE NO LONGER       *
C  REQUIRED).   THE PRIMED COMPONENTS OF THE PROJECTILE VELOCITY ARE     *
C  USED TO DETERMINE THE THREE FACES OF THE TARGET EXPOSED TO THE IN-    *
C  COMING ROUND, AS IT IS IN RELATION TO THE TARGET.                     *
C  CINTRG - TRANSFORMS ANY VECTOR IN INERTIAL SYSTEM TO TARGET SYSTEM    *
C  VTT - VELOCITY VECTOR OF TARGET IN INERTIAL FRAME (VTX,VTY,VTZ)       *
C  VB - VELOCITY VECTOR OF BULLET IN INERTIAL FRAME (VBX,VBY,VBZ)        *
C  SIGN CONVENTION : AZIMUTH - POSITIVE TO THE TARGET'S RIGHT SIDE       *
C                    ELEVATION - POSITIVE UP                             *
C  TARGET COORDINATE SYSTEM - X-AXIS POSITIVE THROUGH THE NOSE           *
C                             Y-AXIS POSITIVE THROUGH THE RIGHT WING     *
C                             Z-AXIS POSITIVE THROUGH THE ROTOR SHAFT    *
C  FACE CONVENTION : 1 = FRONT, 2 = RIGHT SIDE, 3 = REAR, 4 = LEFT       *
C                    SIDE, 5 = TOP, 6 = BOTTOM                           *
C
C****************************************************************************
C
      SUBROUTINE TRGATD(VTT,VB,VS)
      DIMENSION CINTRG(3,3),VTTPRM(3),VBPRIM(3),FACE(6),VTT(3),VB(3)
      COMMON /TRGVA/ VAREA(6,7,4),VV(7),NARMT,IKILL,VA(4),PARAD,PKHX
      COMMON /TARGET/ XT,YT,ZT,VTM,SIGT,GAMT,ALT,RHOT,PSIT,THETAT,
     +                PHIT,ALFT
      COMMON /CNVRSN/ RTD,DTR,GEE,KNTMPS
C
      C1 = COS(PSIT)
      S1 = SIN(PSIT)
      C2 = COS(THETAT)
      S2 = SIN(THETAT)
      C3 = COS(PHIT)
      S3 = SIN(PHIT)
      CINTRG(1,1) = C1*C2
      CINTRG(1,2) = S1*C2
      CINTRG(1,3) = -S2
      CINTRG(2,1) = C1*S2*S3-S1*C3
      CINTRG(2,2) = S1*S2*S3+C1*C3
      CINTRG(2,3) = C2*S3
      CINTRG(3,1) = C1*S2*C3+S1*S3
      CINTRG(3,2) = -C1*S3+S1*S2*C3
      CINTRG(3,3) = C2*C3
      CALL MXMULT(CINTRG,VTT,VTTPRM,3,3,1)
      CALL MXMULT(CINTRG,VB,VBPRIM,3,3,1)
      DO 180 ICOMP=1,4
      IF (VS .GT. VV(1)) GO TO 110
      DO 100 I=1,6
      FACE(I) = VAREA(I,1,ICOMP)*VS/VV(1)
  100 CONTINUE
      GO TO 170
  110 K=2
  120 IF (VS .GT. VV(K)) GO TO 140
      DO 130 I=1,6
      FACE(I) = VAREA(I,K-1,ICOMP) + (VS-VV(K-1))*(VAREA(I,K,ICOMP)-
     +          VAREA(I,K-1,ICOMP))/(VV(K)-VV(K-1))
  130 CONTINUE
      GO TO 170
  140 IF (K .EQ. 7) GO TO 150
      K = K+1
      GO TO 120
  150 DO 160 I=1,6
      FACE(I) = VAREA(I,7,ICOMP)
  160 CONTINUE
  170 CONTINUE
      AZB = 90.
      IF (ABS(VBPRIM(1)) .LT. .00001) GO TO 777
      AZB = ATAN(ABS(VBPRIM(2)/VBPRIM(1)))*RTD
  777 ELB = 90.
      IF (VBPRIM(3) .LT. 0.) ELB = -90.
      A = SQRT(VBPRIM(1)**2 + VBPRIM(2)**2)
      IF (A .LT. .00001) GO TO 778
      ELB = ATAN(VBPRIM(3)/A)*RTD
```

Figure 2 SUBROUTINES "TRGATD" & "TRGINP"

```
  778 IF (VBPRIM(1) .LT. 0. .AND. VBPRIM(2) .LT. 0.) AZB = 180. + AZB
      IF (VBPRIM(1) .LT. 0. .AND. VBPRIM(2) .GE. 0.) AZB = 180. - AZB
      IF (VBPRIM(1) .GE. 0. .AND. VBPRIM(2) .LT. 0.) AZB = 360. - AZB
      IF (AZB .GE. 0. .AND. AZB .LE. 90.) THEN
                          S1 = FACE(3)*COS(AZB*DTR)
                          S2 = FACE(4)*SIN(AZB*DTR)
      ELSEIF (AZB .GE. 90. .AND. AZB .LE. 180.) THEN
                          S1 = FACE(4)*COS((AZB-90.)*DTR)
                          S2 = FACE(1)*SIN((AZB-90.)*DTR)
      ELSEIF (AZB .GE. 180. .AND. AZB .LE. 270.) THEN
                          S1 = FACE(1)*COS((AZB-180.)*DTR)
                          S2 = FACE(2)*SIN((AZB-180.)*DTR)
      ELSEIF (AZB .GE. 270. .AND. AZB .LE. 360.) THEN
                          S1 = FACE(2)*COS((AZB-270.)*DTR)
                          S2 = FACE(3)*SIN((AZB-270.)*DTR)
      ENDIF
      IF (ELB .GE. 0. .AND. ELB .LE. 90.) THEN
                          S3 = FACE(6)*SIN(ABS(ELB*DTR))
      ELSEIF (ELB .GE. -90. .AND. ELB .LE. 0.) THEN
                          S3 = FACE(5)*SIN(ABS(ELB*DTR))
      ENDIF
      S1 = S1*COS(ELB*DTR)
      S2 = S2*COS(ELB*DTR)
      IF (NARMT .EQ. 0) VA(ICOMP) = S1 + S2 + S3
      IF (NARMT .EQ. 1) VA(ICOMP) = SQRT( S1**2 + S2**2 + S3**2)
  180 CONTINUE
      RETURN
      END
C
C*********************************************************************
C
C         NO. 62
C
C   THIS SUBROUTINE INPUTS VULNERABILITY TABLES IN THE EVADE II      *
C   FORMAT -- I=1,6 CARDINAL FACES, J=1,7 STRIKING VELOCITIES, K=1,4  *
C   DAMAGE CONDITIONS WITH PRESENTED AREA OF THE TARGET BEING THE     *
C   FIRST GROUP INPUT. PRESENTED AREAS AND VULNERABLE AREAS ARE INPUT *
C   IN METERS**2, STRIKING VELOCITIES IN METERS/SEC, AND 'NARMT' IS   *
C   ZERO FOR THE SHOEBOX REPRESENTATION OF THE TARGET, AND ONE FOR    *
C   THE ARMITAGE ELLIPSOID REPRESENTATION.  'IKILL' DETERMINES THE    *
C   DEFINITION OF A KILL -- WHEN 'IKILL' EQUALS ONE, ATTRITION VULNER-*
C   ABLE AREA ONLY IS USED, WHEN IT EQUALS TWO, ATTRITION AND FORCED  *
C   LANDING, AND WHEN IT EQUALS THREE, ATTRITION, FORCED LANDING, AND *
C   MISSION ABORT.  THIS FLAG IS USED IN SUBROUTINE PROBKL.           *
C
C*********************************************************************
C
      SUBROUTINE TRGINP
      COMMON /TRGVA/ VAREA(6,7,4),VV(7),NARMT,IKILL,VA(4),PARAD,PKHX
      PKHX=0.0
      DO 50 K=1,4
      DO 40 J=1,7
      READ(9,190) (VAREA(I,J,K),I=1,6)
   40 CONTINUE
   50 CONTINUE
      READ(9,200) (VV(I),I=1,7)
      READ(9,210) NARMT,IKILL
  190 FORMAT(6F10.2)
  200 FORMAT(7F10.1)
  210 FORMAT(2I10)
      RETURN
      END
```

Figure 2 (Continued)

Reference System (HARS). These additional measurements included ownship roll, pitch and yaw rates, pitch and yaw accelerations, and body accelerations with respect to the earth in the aircraft coordinate system. As with the existing error models, the actual values, based on orientation of the attacker, are corrupted by instrument biases and accuracies taken from the procurement specifications for the HARS. The linear prediction algorithm was included immediately following execution of the filter (instead of using two separate subroutines) with the estimated, predicted target position passed directly to the airborne ballistics algorithm.

The COBRA's FCC utilizes a 4-state Kalman filter which estimates target range and three components of target velocity in the sight-line coordinate system. No information, beyond that provided by the sensors already modelled, needed to be derived, so the algorithm was programmed directly into the model. Again, the linear prediction algorithm was connected to the filter with its output passed to the airborne ballistics routine.

The reason for programming the existing filters was to provide insight into the increase in effectiveness of the ATA equations over the current equations, developed primarily for air-to-ground, area suppression operations. The ATA equations have been proposed to be one of the product improvements to the APACHE and COBRA to provide a near-term counter-air capability. Figures 3 and 4 show results of several computer runs completed for this study. Coding for the filters cannot be shown due to company proprietary information contained in the mathematical formulations.

D. As described above, the original model used a quandratic predition algorithm to estimate the future target position. This position is derived using the seven state estimates from the Kalman filter (relative range and




APACHE FIRE CONTROL EQUATIONS COMPARISON

Figure 3




TARGET IN 1 G TURN AND DIVE AT 100 KNOTS
ATTACKER IN CONSTANT 100 KNOT TAIL CHASE

COBRA FIRE CONTROL EQUATIONS COMPARISON

Figure 4

three components of both velocity and acceleration in the sight-line coordinate system) together with the time-of-flight (TOF) of the bullet in the following format:

PREDICTED TARGET RANGE = RANGE + (TOF)(VELOCITY) + $(TOF)^2$ (ACCELERATION) A linear model was programmed, similar to those used in the current filters described above, to study the effect of the prediction method on probability of hit using the ATA equations. The algorithm uses the same format as the quadratic model, but leaves off the acceleration term, as follows:

PREDICTED TARGET RANGE = RANGE + (TOF)(VELOCITY)

As can be expected, the quadratic predictor does much better in the high "g" turning and jinking maneuvers, while in the more benign scenarios, where there are little accelerations, the linear algorithm provides an almost equal estimate because the acceleration term has been dropped.

E. Another deficiency in the original version centered on running the model through one iteration instead of several. This was an unreasonable approach given the amount of randomness in the model. Several runs were made using different random number generator seeds, resulting in wide variations in number of hits per burst and associated kill probability. Therefore, coding was added so that any number of runs could be made at one time with appropriate statistics computed, such as average hits and probability of kill per burst, as well as standard deviations. The model now runs 20 iterations, since standard deviations for most scenarios tend to settle in this time frame, but it can easily be changed. Minor modifications were made to the main program at the beginning of the simulation loop to rewind files for output purposes and to

run to 20 iterations with a new random number generator seed. Subroutine AVG, called at the end of the main program after the appropriate data had been accumulated, computes and outputs the statistics (Figure 5).

F. The model now has the ability to access a Tektronix, Inc. graphics package called PLOT-10, resident on AVSOCM's IBM mainframe computer. The user has the option to run the graphics program, which plots several charts that display Kalman filter performance. The three graphs (also done by TSC for their final report) show magnitude of range, velocity and acceleration errors and standard deviations over the 12 second engagement. They can be used to study how accurate the filter's estimates are, settling time of those estimates and effects of sensor input rates and accuracies on the estimates. The FORTRAN PLOT-10 program, PLOT FORTRAN, is called by the master executive file after the main program is done and has output the variables to be plotted to a separate data file (Figure 6).

G. Finally, many more engagement input files have been created to study key fire control system parameters. These engagements include everything from both ownship and target hovering (to baseline the new filters programmed into the model), to both aircraft maneuvering. Most of the engagements were developed for LHX purposes to examine the requirements of hit and kill stated in the LHX airframe request for proposals (RFP) for the turreted gun system. Also, runs were made to study the trade-offs between using 20MM or 30MM cannons in ATA combat, the main trade being between aerodynamic qualities of the round (which translates into time-of-flight) and lethality of it on the target. Several of the engagements are now contained in LHX specifications for modelling by the two contractor teams to validate their fire control system performance abilities.

```
C
C*************************************************************************
C
C         NO. 14
C
C     THIS SUBROUTINE COMPUTES THE AVERAGE, STANDARD DEVIATION AND
C     UPPER AND LOWER BOUNDS OF THE NUMBER OF HITS AND CUMULATIVE
C     PROBABILITY OF KILL FOR THE TWO BURSTS AFTER 20 REPLICATIONS.
C
C*************************************************************************
      SUBROUTINE AVG(NENGAG,KBUL30)
      REAL KILL1(20),KILL2(20),LOWER(4)
      COMMON /MONCAR/ HITS1(20),KILL1,HITS2(20),KILL2
      DIMENSION SUMM(4),SUMSQ(4),XBAR(4),SQSUM(4),VAR(4),SDEV(4),
     +          UPPER(4)
      TEE = 2.093
      SAMPLE = 20.
      DO 10 K = 1,4
      SUMM(K) = 0.0
      SUMSQ(K) = 0.0
      XBAR(K) = 0.0
      SQSUM(K) = 0.0
      VAR(K) = 0.0
      SDEV(K) = 0.0
      UPPER(K) = 0.0
      LOWER(K) = 0.0
   10 CONTINUE
      DO 20 I=1,20
      SUMM(1) = SUMM(1) + HITS1(I)
      SUMM(2) = SUMM(2) + KILL1(I)
      SUMM(3) = SUMM(3) + HITS2(I)
      SUMM(4) = SUMM(4) + KILL2(I)
      SUMSQ(1) = SUMSQ(1) + HITS1(I)*HITS1(I)
      SUMSQ(2) = SUMSQ(2) + KILL1(I)*KILL1(I)
      SUMSQ(3) = SUMSQ(3) + HITS2(I)*HITS2(I)
      SUMSQ(4) = SUMSQ(4) + KILL2(I)*KILL2(I)
   20 CONTINUE
      DO 30 J=1,4
      XBAR(J) = SUMM(J)/SAMPLE
      SQSUM(J) = SUMM(J)*SUMM(J)
      VAR(J) = (1./(SAMPLE*(SAMPLE-1)))*(SAMPLE*SUMSQ(J)-SQSUM(J))
      SDEV(J) = SQRT(VAR(J))
      UPPER(J) = XBAR(J) +(TEE*SDEV(J)/SQRT(SAMPLE))
      LOWER(J) = XBAR(J) -(TEE*SDEV(J)/SQRT(SAMPLE))
   30 CONTINUE
      WRITE(8,100) NENGAG
  100 FORMAT(/3X,'ENGAGEMENT NUMBER',I4,//)
      IF (KBUL30 .EQ. 0) THEN
         WRITE(8,200)
  200    FORMAT(3X,'BULLET USED IS 20MM HEI'//)
      ELSEIF (KBUL30 .EQ. 2) THEN
         WRITE(8,300)
  300    FORMAT(3X,'BULLET USED IS 20MM PIE'//)
      ELSEIF (KBUL30 .EQ. 1) THEN
         WRITE(8,400)
  400    FORMAT(3X,'BULLET USED IS 30MM HEDP'//)
      ENDIF
      WRITE(8,500)
  500 FORMAT(/3X,'STATISTICS AFTER 20 REPLICATIONS :'//
     + 15X,'AVERAGE',10X,'BOUNDS FOR 95 %',23X,'BOUNDS FOR 95 %'/
     + 17X,'HITS',9X,'CONFIDENCE INTERVAL',4X,'PROBABILITY',4X,
     + 'CONFIDENCE INTERVAL'/
     + 13X,'(STAND. DEV.)',6X,'LOWER',5X,'UPPER',8X,'OF KILL',8X,
     + 'LOWER',5X,'UPPER'/
     + 13X,'-------------',6X,'-----',5X,'-----',8X,'-------',8X,
     + '-----',5X,'-----'/)
      WRITE(8,600) XBAR(1),LOWER(1),UPPER(1),XBAR(2),LOWER(2),UPPER(2),
     +   SDEV(1),XBAR(3),LOWER(3),UPPER(3),XBAR(4),LOWER(4),UPPER(4),
     +   SDEV(3)
  600 FORMAT(1X,'BURST 1',9X,F5.2,10X,F5.2,5X,F5.2,9X,F4.3,10X,F4.3,6X,
     + F4.3,/16X,'(',F5.2,')',//
     + 1X,'BURST 2',9X,F5.2,10X,F5.2,5X,F5.2,9X,F4.3,10X,F4.3,6X,F4.3,/
     + 16X,'(',F5.2,')')
      RETURN
      END
```

SUBROUTINE "AVG"
Figure 5

```
      DIMENSION RNGERR(50), RNGSD(50),VELERR(50),VELSD(50),ACCERR(50),
     +          ACCSD(50),XDATA(14)
      DATA XDATA/13.,0.,1.,2.,3.,4.,5.,6.,7.,8.,9.,10.,11.,12./
      RNGERR(1)=13.
      RNGSD(1)=13.
      VELERR(1)=13.
      VELSD(1)=13.
      ACCERR(1)=13.
      ACCSD(1)=13.
      REWIND 10
      DO 2 I=2,14
      READ (10,1) RNGERR(I),RNGSD(I)
      READ (10,1) VELERR(I),VELSD(I)
      READ (10,1) ACCERR(I),ACCSD(I)
    1 FORMAT (2F10.5)
    2 CONTINUE
      CALL INITT(120)
      BMIN=-6.00000
      BMAX=6.00000
      CALL BINITT
      CALL DLIMY(BMIN,BMAX)
      CALL LINE(0)
      CALL CHECK(XDATA,RNGERR)
      CALL DSPLAY(XDATA,RNGERR)
      CALL LINE(1)
      CALL CPLOT(XDATA,RNGSD)
      CALL MOVABS(400,750)
      CALL AOUTST(29,'KALMAN FILTER CHARACTERISTICS')
      CALL MOVABS(400,735)
      CALL AOUTST(29,'            RANGE              ')
      CALL MOVABS(400,720)
      CALL AOUTST(29,'   ENGAGEMENT NUMBER           ')
      CALL MOVABS(485,35)
      CALL AOUTST(10,'TIME (SEC)')
      CALL MOVABS(20,410)
      CALL AOUTST(6,'METERS')
      CALL MOVABS(650,65)
      CALL AOUTST(15,'   MAGNITUDE OF')
      CALL MOVABS(650,52)
      CALL AOUTST(15,'    RANGE ERROR')
      CALL DSHABS(800,52,0)
      CALL MOVABS(650,26)
      CALL AOUTST(15,'   MAGNITUDE OF')
      CALL MOVABS(650,13)
      CALL AOUTST(15,'RANGE STD. DEV.')
      CALL DSHABS(800,13,1)
      CALL TINPUT(ICHAR)
      BMIN=0.00000
      BMAX=50.00000
      CALL BINITT
      CALL NEWPAG
      CALL DLIMY(BMIN,BMAX)
      CALL LINE(0)
      CALL CHECK(XDATA,VELERR)
      CALL DSPLAY(XDATA,VELERR)
```

PLOT FORTRAN PROGRAM

Figure 6

```
      CALL LINE(1)                                                      PL000560
      CALL CPLOT(XDATA,VELSD)                                           PL000570
      CALL MOVABS(400,750)                                              PL000580
      CALL AOUTST(29,'KALMAN FILTER CHARACTERISTICS')                   PL000590
      CALL MOVABS(400,735)                                              PL000600
      CALL AOUTST(29,'         VELOCITY           ')                    PL000610
      CALL MOVABS(400,720)                                              PL000620
      CALL AOUTST(29,'    ENGAGEMENT NUMBER       ')                    PL000630
      CALL MOVABS(485,35)                                               PL000640
      CALL AOUTST(10,'TIME (SEC)')                                      PL000650
      CALL MOVABS(1,410)                                                PL000660
      CALL AOUTST(12,'METERS / SEC')                                    PL000670
      CALL MOVABS(650,65)                                               PL000680
      CALL AOUTST(18,'      MAGNITUDE OF')                              PL000690
      CALL MOVABS(650,52)                                               PL000700
      CALL AOUTST(18,'    VELOCITY ERROR')                              PL000710
      CALL DSHABS(800,52,0)                                             PL000720
      CALL MOVABS(650,26)                                               PL000730
      CALL AOUTST(18,'      MAGNITUDE OF')                              PL000740
      CALL MOVABS(650,13)                                               PL000750
      CALL AOUTST(18,'VELOCITY STD. DEV.')                              PL000760
      CALL DSHABS(800,13,1)                                             PL000770
      CALL TINPUT(ICHAR)                                                PL000780
      BMIN=0.00000                                                      PL000790
      BMAX=40.00000                                                     PL000800
      CALL BINITT                                                       PL000810
      CALL NEWPAG                                                       PL000820
      CALL DLIMY(BMIN,BMAX)                                             PL000830
      CALL LINE(0)                                                      PL000840
      CALL CHECK(XDATA,ACCERR)                                          PL000850
      CALL DSPLAY(XDATA,ACCERR)                                         PL000860
      CALL LINE(1)                                                      PL000870
      CALL CPLOT(XDATA,ACCSD)                                           PL000880
      CALL MOVABS(400,750)                                              PL000890
      CALL AOUTST(29,'KALMAN FILTER CHARACTERISTICS')                   PL000900
      CALL MOVABS(400,735)                                              PL000910
      CALL AOUTST(29,'       ACCELERATION         ')                    PL000920
      CALL MOVABS(400,720)                                              PL000930
      CALL AOUTST(29,'    ENGAGEMENT NUMBER       ')                    PL000940
      CALL MOVABS(485,35)                                               PL000950
      CALL AOUTST(10,'TIME (SEC)')                                      PL000960
      CALL MOVABS(1,410)                                                PL000970
      CALL AOUTST(10,'METERS/SEC')                                      PL000980
      CALL MOVABS(80,415)                                               PL000990
      CALL AOUTST(1,'2')                                                PL001000
      CALL MOVABS(650,65)                                               PL001010
      CALL AOUTST(22,'          MAGNITUDE OF')                          PL001020
      CALL MOVABS(650,52)                                               PL001030
      CALL AOUTST(22,'    ACCELERATION ERROR')                          PL001040
      CALL DSHABS(850,52,0)                                             PL001050
      CALL MOVABS(650,26)                                               PL001060
      CALL AOUTST(22,'          MAGNITUDE OF')                          PL001070
      CALL MOVABS(650,13)                                               PL001080
      CALL AOUTST(22,'ACCELERATION STD. DEV.')                          PL001090
      CALL DSHABS(850,13,1)                                             PL001100
      CALL TINPUT(ICHAR)                                                PL001110
      CALL FINITT(0,700)                                                PL001120
      STOP                                                              PL001130
      END                                                               PL001140
```

Figure 6 (Continued)

## IV. FUTURE MODIFICATIONS

A. Several changes are still planned to be made to the model. TSC has completed the addition of 2.75 inch Hydra 70 flechette warhead rockets to the simulation. Due to the size and run time of the flyout model for the rockets, the number of iterations used currently will have to be reduced to save computer time. Also, not all necessary lethality data from field testing has been generated by BRL and AMSAA at this writing, which limits the end-game analysis.

B. The addition of a dueling module, so that the target aircraft can return fire at the attacker, has been discussed as another modification. Methodology similar to AMSAA's Helicopter Air-to-Air Combat Simulation (HATACS) could be adopted for usage in TSC's model.

C. The ability for the model to access actual flight test data has already been discussed and is being looked at by both ARDEC and AVSCOM. This capability would be helpful during flight testing of the ATA equations in the COBRA testbed to exploit as a potential debugging tool, while concurrently validating the model. Also, flight paths from the Air-to-Air Combat Tests I-IV (AACT) at Paxtuxent River sponsored by AATD could be modelled.

D. Several minor modifications have also been planned. These include adding other filters to study in ATA combat. Error budgets are currently being performed for the COBRA and APACHE gun systems, allowing for validation of the numbers used to represent sensor measurement errors in the model. The turret module needs to be upgraded to increase accuracy of actual turret movement and firing. Finally, addition of new sensors, such as a millimeter-wave

radar, closed-loop fire control system radar or muzzle velocity sensor can be accomplished with appropriate error inputs.

## CONCLUSIONS

The TSC AIRTOAIR model is beginning to become a widely accepted tool, by both contractor and government analysts, for studying cannon/rocket effectiveness in ATA combat. It is hoped that this report will allow all users of the model to make any of these upgrades they feel as necessary. Also, maintaining a constant version of the model would be favorable, with all users providing documentation of changes made for the widest dissemination as possible.

## REFERENCES

1. Breaux, Harold J., A Methodology for the Development of Fire Control Equations for Guns and Rockets Fired from Aircraft, USA BRL Interim Memorandum Report No. 827, October 1984.

2. TSC, Attack Helicopter Air-to-Air Fire Control System Simulation, Final Report (Part 1 - Technical) submitted to the USA ARDEC under contract No. DAAK10-82-C-0055, August 1983.

LIST OF ABBREVIATIONS, ACRONYMS & SYMBOLS

| | |
|---|---|
| PIE | Pyrotechnically initiated explosive |
| ATA | Air-to-air |
| TSC | Teledyne Systems Company |
| USAAVSCOM | U.S. Army Aviation Systems Command |
| AATD | Aviation Applied Technology Directorate |
| ARDEC | Armament Research, Development and Engineering Center |
| FCC | Fire Control Computer |
| BRL | Ballistics Research Laboratory |
| DOF | Degree-of-freedom |
| LHX | Light Helicopter Family |
| HEDP | High explosive, dual purpose |
| AMSAA | Army Materiel Systems Analysis Activity |
| HARS | Heading Attitude Reference System |
| TOF | Time-of-flight |
| RFP | Request for proposals |
| HATACS | Helicopter Air-to-Air Combat Simulation |
| AACT | Air-to-Air Combat Tests |

END
DATE
FILMED
MARCH
1988
DTIC